

# **Applicazioni con EXCEL alle decisioni finanziarie**

**Appunti per il corso di Metodi decisionali per l'azienda B  
Corso di laurea in statistica e informatica per la gestione delle imprese  
a.a. 2001-2002**

**Stefania Funari**

*Dipartimento di Matematica Applicata  
Università Ca' Foscari di Venezia*

e-mail: [funari@unive.it](mailto:funari@unive.it)  
<http://helios.unive.it/~funari>



## **INDICE**

### ***Premessa 3***

### ***SCelta DI PROGETTI FINANZIARI 4***

- 1. VALORE ATTUALE NETTO 4***
  - 1.1 Definizione 4*
  - 1.2 La funzione VAN 4*
  - 1.3 La funzione VA 7*
  - 1.4 Uso del comando Ricerca obiettivo 9*
- 2. TASSO INTERNO DI RENDIMENTO 13***
  - 2.1 Definizione 13*
  - 2.2 La funzione TIR.COST 13*
  - 2.3 Uso del comando Ricerca obiettivo 15*
- 3 CRITERIO T.R.M. 17***
  - 3.1 Descrizione 17*
  - 3.2 Uso della funzione SE 18*

### ***VALUTAZIONE DI RENDITE 21***

- 1. Montante di rendite periodiche posticipate 21*
- 2. Montante di rendite periodiche anticipate 26*
- 3. valore attuale di rendite periodiche 27*

### ***PIANO DI AMMORTAMENTO DI UN DEBITO 29***

- 1. AMMORTAMENTO ITALIANO 30*
- 2. AMMORTAMENTO FRANCESE 34*

### ***ANALISI PER IPOTESI: USO DI TABELLE DI DATI 39***

- 1. Esempio: una variabile di input 39*
- 2. Esempio: due variabili di input 41*

## PREMESSA

Si sono raccolte in questo rapporto didattico alcune note, scritte per il corso di *Metodi decisionali per l'azienda B*, corso di laurea in *Statistica e Informatica per la gestione delle imprese*, sull'uso di Excel ai fini della risoluzione di alcuni problemi di matematica finanziaria.

L'utilizzo di un foglio elettronico può essere di ausilio nella formulazione e risoluzione di alcuni semplici problemi di matematica finanziaria; è possibile ad esempio, usando Excel, calcolare il valore attuale di una rendita, oppure costruire il piano di ammortamento di un debito, oppure selezionare, in base ad un determinato criterio di scelta, il miglior progetto di investimento. Excel infatti permette, una volta impostato correttamente un problema in un foglio di lavoro, di eseguire direttamente le operazioni richieste per la risoluzione del problema facendo anche uso di funzioni finanziarie già predefinite. Inoltre Excel fornisce alcune funzioni avanzate per eseguire *analisi per ipotesi* che consentono di verificare l'impatto sul modello decisionale di cambiamenti nel valore di alcune variabili. Per risolvere problemi di finanza più complessi, che non saranno trattati in questa sede, è poi possibile utilizzare la programmazione in ambiente Windows e creare macro di Excel con Visual Basic.

Il presente lavoro è articolato in diverse parti ciascuna delle quali è dedicata all'analisi di specifici problemi di matematica finanziaria che sono risolti impiegando *Microsoft Excel 2000*. Le pagine seguenti presuppongono la conoscenza delle principali operazioni elementari in ambiente Excel, come ad esempio l'apertura, il salvataggio e la chiusura di un file, la selezione di una o più celle, l'immissione di dati e l'inserimento di formule nelle celle di un foglio di lavoro.

# SCELTA DI PROGETTI FINANZIARI

## 1. VALORE ATTUALE NETTO

### 1.1 Definizione

Dato un progetto finanziario caratterizzato da un insieme di importi  $S_0, S_1, \dots, S_k, \dots, S_n$ , che si verificano alle epoche temporali  $0, 1, \dots, k, \dots, n$  si definisce valore attuale netto del progetto la somma dei valori attuali calcolati in  $t = 0$  degli importi  $S_k$ , considerato un regime di capitalizzazione composta al tasso di interesse  $i$

$$V(i) = \sum_{k=0}^n S_k (1+i)^{-k} \quad (1)$$

### 1.2 La funzione VAN

Per calcolare il valore attuale netto di un progetto si può utilizzare la funzione finanziaria VAN di Excel. La funzione si presenta nella forma

$$\text{VAN}(\text{tasso\_int}; \text{valore1}; \text{valore2}; \dots; \text{valoren})$$

dove gli argomenti della funzione hanno il seguente significato:

*tasso\_int*                      Tasso di interesse relativo all'unità temporale considerata.  
*valore1, ..., valoren*        Importi periodici  $S_1, \dots, S_n$  riscossi o pagati nelle diverse epoche.

Excel 2000 ammette al massimo 29 argomenti *valore*. É comunque possibile considerare un qualsiasi numero di valori nella funzione utilizzando una matrice come argomento.

Gli importi periodici  $S_k$  sono considerati equamente distribuiti alla fine di ciascuna epoca temporale. Occorre fare attenzione al primo importo ( $S_0$ ). Esso non viene incluso come argomento della funzione VAN, per cui per ottenere il valore attuale netto (1) occorre sommare (algebricamente) il risultato della funzione VAN con l'importo  $S_0$ . Gli esempi che seguono costituiscono applicazioni della funzione VAN.

### **Esempio**

*Si vuole calcolare, utilizzando un tasso di interesse del 10%, il valore attuale netto di un investimento finanziario che prevede un esborso iniziale di 10000 Euro, la*

riscossione di 7000 Euro alla fine del primo anno e la riscossione di 6000 Euro alla fine del secondo anno.

I dati del problema sono descritti nelle seguenti celle di un foglio di lavoro Excel, come evidenziato nella figura 1:


cella B4	-10000	importo alla scadenza 0
cella B5	7000	importo alla scadenza 1
cella B6	6000	importo alla scadenza 2
cella B8	0.1	tasso di interesse

Per calcolare il valore attuale netto del progetto di investimento posizionarsi nella cella B9, scrivere

$$=B4+VAN(B8;B5;B6)$$

e premere INVIO. Il risultato è 1322.314.

figura 1



### Esempio

Si vuole calcolare, utilizzando un tasso di interesse del 6%, il valore attuale netto al tempo  $t=0$  di un progetto i cui movimenti di cassa sono i seguenti:

<i>epoca</i>	<i>importo</i>
0	0
1	50
2	-20
3	-10
4	30


Vogliamo utilizzare la funzione VAN di Excel. In un foglio di lavoro come quello descritto in figura 2 le celle da B4 a B7 contengono gli importi che descrivono il progetto finanziario a partire dall'epoca 1; la cella B9 contiene il valore del tasso di interesse.

Per calcolare il valore attuale netto posizionarsi nella cella B10, scrivere

$$=VAN(B9;B4:B7)$$

e premere INVIO. Il risultato è 44.736.

figura 2



Se non si fosse utilizzata la funzione finanziaria di Excel si sarebbe comunque potuto calcolare il valore attuale netto del progetto digitando direttamente la formula di calcolo. Secondo la (1) il valore attuale netto del progetto è calcolato come

$$V = \frac{50}{(1+0.06)} - \frac{20}{(1+0.06)^2} - \frac{10}{(1+0.06)^3} + \frac{30}{(1+0.06)^4}$$

Si può quindi inserire nella cella B10 direttamente la formula:

$$=(B4/(1+B9))+(B5/(1+B9)^2)+(B6/(1+B9)^3)+(B7/(1+B9)^4)$$

e premere INVIO. Il risultato è 44.736.

In questo modo si usano molti più caratteri di quelli impiegati con la funzione finanziaria di Excel.

### **Esempio**

*Scegliere, utilizzando il criterio del valore attuale netto al tasso del 5%, il progetto più conveniente fra i progetti A e B i cui movimenti di cassa sono i seguenti:*

<b>PROGETTO A</b>		<b>PROGETTO B</b>	
<i>epoca</i>	<i>importo</i>	<i>epoca</i>	<i>importo</i>
0	-20	0	0
1	5	1	28
2	40	2	-30
3	-8	3	15

Per calcolare il valore attuale netto del progetto A posizionarsi in una cella di un foglio di lavoro Excel, digitare

$$= - 20+VAN(0.05;5;40;-8)$$

e premere INVIO. Si ottiene come risultato: 14.132.

Per calcolare il valore attuale netto del progetto B, posizionarsi in un'altra cella del foglio di lavoro, digitare

$$= VAN(0.05;28;-30;15)$$

e premere INVIO. Si ottiene come risultato: 12.413.

Poiché il valore attuale netto di A è maggiore del valore attuale netto di B si sceglie come progetto più conveniente il progetto A.

## **1.3 La funzione VA**

Qualora il progetto finanziario sia caratterizzato da poste  $S_k$  uguali ad un comune importo  $S$  indipendentemente dall'epoca di riferimento, può essere conveniente utilizzare la funzione finanziaria VA di Excel.

Essa permette anche di considerare la situazione in cui gli importi siano pagati o riscossi anticipatamente, cioè all'inizio di ciascuna epoca di riferimento.

Inoltre la funzione VA, pur prevedendo la situazione di uguaglianza dei movimenti di cassa alle diverse scadenze, permette di considerare la possibilità che all'epoca finale vi sia un'eventuale pagamento (o riscossione)  $F$ , aggiuntivo rispetto ai movimenti del progetto.

La funzione si presenta nella forma

$$VA(\text{tasso\_int};\text{periodi};\text{pagamento};\text{valore\_futuro};\text{tipo})$$

dove gli argomenti della funzione hanno il seguente significato

<i>tasso_int.</i>	Tasso di interesse relativo all'unità temporale considerata.
<i>periodi</i>	Numero totale dei pagamenti.
<i>pagamento</i>	Pagamento effettuato in ciascun periodo.
<i>valore_futuro</i>	Pagamento addizionale <i>F</i> effettuato nell'ultima epoca, che si aggiunge all'ultimo pagamento.
<i>tipo</i>	Gestisce il tipo di convenzione usata per i pagamenti; se è uguale a 0 (oppure omissso) indica che gli importi sono posticipati; se è uguale ad 1 indica che gli importi sono anticipati.

Potrebbe accadere che l'argomento *valore\_futuro* manchi, oppure che ci sia *valore\_futuro*, ma manchi l'argomento *pagamento*, oppure che vi siano entrambi. Ciascuna situazione corrisponde ad una particolare struttura finanziaria del progetto.

Prima di passare a qualche esempio, facciamo notare che occorre molta attenzione nell'uso delle funzioni finanziarie di Excel. La funzione VA, ad esempio, fornisce un valore di segno opposto rispetto a quello dei flussi di cassa *S*, per cui il valore attuale sarà calcolato come l'opposto del risultato della funzione VA. I seguenti esempi chiariscono questa questione.

### **Esempio**

*Si vuole calcolare, utilizzando un tasso di interesse dell'8%, il valore attuale, al tempo  $t=0$ , di un progetto che prevede di riscuotere una somma di 850 Euro alla fine di ogni anno, per 6 anni.*

Per calcolare il valore attuale basta posizionarsi in una qualsiasi cella di un foglio di lavoro, scrivere

$$= -VA(0.08;6;850)$$

e premere INVIO. Il risultato è: 3929.448.

La funzione VA non possiede in questo caso né l'argomento *valore\_futuro* né l'argomento *tipo*. L'omissione di quest'ultimo indica che si stanno considerando importi posticipati.

La funzione restituisce un valore negativo

$$VA(0.08;6;850) = - 3929.448$$

La motivazione è la seguente: si dovrebbero spendere 3929.448 Euro per ricevere 850 Euro alla fine di ogni anno, per i prossimi 6 anni. Se quindi al tempo  $t=0$  si paga una somma minore di 3929 il progetto risulterà conveniente.

### **Esempio**

Si vuole calcolare, utilizzando un tasso di interesse dell'8%, il valore attuale, al tempo  $t=0$ , di un progetto che prevede di riscuotere una somma di 2500 Euro alla fine del sesto anno.

Per calcolare il valore attuale basta posizionarsi in una qualsiasi cella di un foglio di lavoro, scrivere

$$= -VA(0.08;6;;2500)$$

e premere INVIO. Il risultato è: 1575.424.

In questo esempio non si prevedono le riscossioni periodiche ma solamente un unico prelievo finale. La funzione VA possiede in questo caso l'argomento *valore\_futuro* e non l'argomento *pagamento*. Si noti il punto e virgola aggiuntivo che agisce da separatore per l'argomento *pagamento* non impiegato.

## **1.4 Uso del comando Ricerca obiettivo**

Il comando *Ricerca obiettivo* del menu *Strumenti* permette di trovare il valore di una certa variabile in modo da raggiungere un prefissato obiettivo.

Tale comando può essere utilizzato, ad esempio, per trovare il tasso di interesse oppure il costo iniziale del progetto che permette di ottenere un determinato valore attuale netto.

*Ricerca obiettivo* è quindi utile quando si conosca un risultato e si vuole cercare il valore di una variabile che lo ha prodotto.

### Come funziona Ricerca obiettivo


Quando dal menu *Strumenti* si seleziona *Ricerca obiettivo* compare una finestra di dialogo con tre voci: *Imposta la cella*, *Al valore*, *Cambiando la cella* (figura 3).

*Imposta la cella*: rappresenta il riferimento alla "cella obiettivo"; la cella obiettivo deve contenere una formula (e non un valore).

*Al valore*: rappresenta il valore che vogliamo assuma la cella obiettivo; si richiede in questa casella l'inserimento di un valore.

*Cambiando la cella*: rappresenta il riferimento alla "cella soluzione"; la cella soluzione deve contenere un valore (e non una formula).

figura 3



Ricerca obiettivo opera seguendo un procedimento iterativo di ricerca della soluzione: comincia ponendo un valore nella “cella soluzione” (indicata nella casella *Cambiando la cella*) e confronta il valore calcolato nella cella “obiettivo” (indicata nella casella *Imposta cella*) con il valore indicato nella casella *Al valore*. Se i due valori non coincidono, sceglie un altro valore per la “cella soluzione” e ripete il procedimento.

Per default Ricerca obiettivo compie un numero massimo di 100 iterazioni, fermando il procedimento quando la soluzione differisce al più di 0.001 dal valore cercato. È possibile comunque cambiare le impostazioni, per aumentare il numero di iterazioni oppure per diminuire lo scarto massimo (si seleziona *Strumenti, Opzioni* e si attiva la scheda *Calcolo*).

A volte Ricerca obiettivo potrebbe non trovare una soluzione o perché una soluzione non esiste o perché i dati non sono corretti. Si riceve in questo caso un messaggio d'errore.


### **Esempio**

*Dato un investimento finanziario che prevede un esborso iniziale di 10000 Euro, la riscossione di 7000 Euro alla fine del primo anno e la riscossione di 6000 Euro alla fine del secondo anno, si vuole determinare il tasso di interesse che dà luogo ad un valore attuale netto dell'investimento di 1950 Euro.*

I dati caratterizzanti l'investimento sono uguali a quelli del primo esempio del par. 1.2. In questo caso si conosce il valore attuale netto dell'investimento e si vuole trovare il valore del tasso di interesse che lo giustifica.

Si parte dal foglio di calcolo costruito come in figura 4. Le celle da B4 a B6 contengono i dati relativi al progetto di investimento; la cella B8 contiene informazioni sul tasso di interesse, mentre in B9 si ha la formula per il calcolo del valore attuale netto.

figura 4



Per risolvere il problema si proceda come segue.

- Dal menu *Strumenti* scegliere il comando *Ricerca obiettivo* (appare la *finestra di dialogo Ricerca obiettivo*).
- Nella casella *Imposta la cella* della finestra di dialogo scrivere B9
- Nella casella *Al valore* digitare il valore attuale netto che si desidera ottenere: in questo caso 1950
- Nella casella *Cambiando la cella* digitare B8 (il riferimento al tasso di interesse incognito).
- Confermare con OK. Appare la finestra di dialogo *Stato ricerca obiettivo* che informa che è stata trovata una soluzione
- Confermare con OK. Il risultato cercato comparirà nella cella B8. Si ottiene un tasso di interesse del 6% (vedere figura 5)

figura 5

The screenshot shows a Microsoft Excel window titled "Microsoft Excel - Van-3bis.xls". The spreadsheet contains the following data:

	A	B	C	D	E	F	G	H
1	<b>soluzione</b>							
2								
3	<b>scadenze</b>	<b>importi</b>						
4	0	-10000						
5	1	7000						
6	2	6000						
7								
8	<b>tasso di interesse</b>	0.06						
9	<b>valore attuale netto</b>	1950						
10								
11								
12								
13								
14								
15								
16								

The status bar at the bottom shows "Pronto" and "NUM".

## 2. TASSO INTERNO DI RENDIMENTO

### 2.1 Definizione

Il tasso interno di rendimento di un progetto finanziario rappresenta il tasso di interesse che rende uguale a zero il valore attuale netto del progetto; si ottiene risolvendo l'equazione

$$V(i) = \sum_{k=0}^n S_k (1+i)^{-k} = 0$$

### 2.2 La funzione TIR.COST

Per calcolare il tasso interno di rendimento di un progetto finanziario si può usare la funzione finanziaria di Excel TIR.COST, che si presenta nella forma

$$= \text{TIR.COST}(\text{valori}; \text{ipotesi})$$

Gli argomenti della funzione hanno il seguente significato:

<i>valori</i>	Sequenza di flussi di cassa.
<i>ipotesi</i>	Tasso con cui Excel inizia il procedimento di calcolo iterativo; se <i>ipotesi</i> viene omissa, Excel inizia con un tasso di interesse uguale a 0.1.

La funzione TIR.COST restituisce il tasso interno di rendimento, inteso come il tasso effettivo relativo alla periodicità dei flussi considerati (se ad es. i flussi sono trimestrali, il valore restituito dalla funzione TIR.COST è il tasso effettivo trimestrale).

È anche possibile usare la funzione finanziaria di Excel TASSO.

La differenza fra le funzioni TIR.COST e TASSO è analoga alle differenze intercorrenti fra le funzioni VAN e VA. La funzione TIR.COST considera importi di capitale che possono essere diversi uno dall'altro, mentre in TASSO gli importi di capitale sono ipotizzati uguali; inoltre la funzione TASSO permette di trattare anche la situazione in cui i movimenti finanziari sono eseguiti in modo anticipato.

#### **Esempio**

*Si vuole calcolare il tasso interno di rendimento dell'investimento finanziario che prevede un esborso iniziale di 10000 Euro, la riscossione di 7000 Euro alla fine del primo anno e la riscossione di 6000 Euro alla fine del secondo anno.*

I dati che caratterizzano il progetto di investimento sono (vedere figura 6)


cella B4	-10000	(importo alla scadenza 0)
cella B5	7000	(importo alla scadenza 1)
cella B6	6000	(importo alla scadenza 2)

Per calcolare il tasso interno di rendimento del progetto selezionare la cella B8, scrivere

$$= \text{TIR.COST}(B4:B6)$$

e premere INVIO. Il risultato è 0.2, cioè un tasso interno di rendimento del 20%.

figura 6



### Esempio

Scegliere, utilizzando il criterio del tasso interno di rendimento il progetto più conveniente fra i progetti A e B i cui movimenti di cassa sono i seguenti:

<b>PROGETTO A</b>	
<i>epoca</i>	<i>importo</i>
0	-13000
1	4000
2	5900
3	1800
4	6000

<b>PROGETTO B</b>	
<i>epoca</i>	<i>importo</i>
0	-28000
1	2000
2	13000
3	9500
4	10500

In un foglio di lavoro come quello rappresentato nella figura 7 immettere nelle celle da B5 a B9 gli importi che descrivono il progetto di investimento A, e nelle celle da C5 a C9 gli importi che descrivono il progetto di investimento B.

figura 7

1	Scelta di investimenti			
2		Movimenti di cassa	Movimenti di cassa	
3	scadenze	PROGETTO A	PROGETTO B	
4				
5	0	-13000	-28000	
6	1	4000	2000	
7	2	5900	13000	
8	3	1800	9500	
9	4	6000	10500	
10				
11		TIR progetto A	TIR progetto B	
12		0.133	0.084	
13				
14				

Per il calcolo del tasso interno di rendimento di A, posizionarsi nella cella B12, scrivere  
 $= \text{TIR.COST} (B5:B9)$

e premere INVIO. Il risultato è 0.133.

Per il calcolo del tasso interno di rendimento di B, posizionarsi nella cella C12, scrivere  
 $= \text{TIR.COST} (C5:C9)$

e premere INVIO. Il risultato è 0.084.

Poiché il tasso interno di rendimento del progetto di investimento A è maggiore del tasso interno di rendimento di B, si sceglie il progetto A.

### 2.3 Uso del comando Ricerca obiettivo

Si utilizza il comando *Ricerca obiettivo* per trovare quale dovrebbe essere l'importo riscosso o pagato ad una determinata epoca del progetto affinché il tasso interno di rendimento del progetto sia pari ad un valore specificato.

#### Esempio

Sia dato un progetto finanziario che prevede un esborso iniziale di X Euro, la riscossione di 3000 Euro alla fine del primo anno e il pagamento di 250 Euro alla fine

del secondo anno. Calcolare l'esborso iniziale  $X$  che permette di ottenere un tasso interno di rendimento del 12%.


Si parte da un foglio di lavoro costruito come in figura 8.

- Selezionare la cella B5, digitare 3000 e premere INVIO.
- Selezionare la cella B6, digitare -250 e premere INVIO.
- Selezionare la cella B4, immettere un qualsiasi numero negativo, ad esempio -1000 e premere INVIO.
- Nella cella B8 scrivere

$$= \text{TIR.COST}(B4:B6)$$

e premere INVIO.

figura 8



- Dal menu *Strumenti* scegliere il comando *Ricerca obiettivo* (appare la *finestra di dialogo Ricerca obiettivo*).
- Nella casella *Imposta la cella* della finestra di dialogo scrivere B8.
- Nella casella *Al valore* digitare il tasso interno di rendimento che si desidera ottenere: in questo caso 0.12.
- Nella casella *Cambiando la cella* digitare B4 (il riferimento al primo esborso incognito).
- Confermare con OK. Appare la finestra di dialogo *Stato ricerca obiettivo* che informa che è stata trovata una soluzione
- Confermare con OK. Il risultato cercato comparirà nella cella B4. Si ottiene  $X = -2479$  (vedere figura 9).

figura 9

	A	B	C	D	E	F	G	H
1	soluzione							
2								
3	scadenze	importi						
4	0	-2479						
5	1	3000						
6	2	-250						
7								
8	TIR	0.120						
9								
10								
11								
12								
13								
14								
15								
16								
17								
18								
19								
20								

### 3 CRITERIO T.R.M.

#### 3.1 Descrizione

Si considera un progetto finanziario che prevede i movimenti di cassa  $S_0, S_1, \dots, S_k, \dots, S_n$ , che si verificano alle scadenze temporali  $0, 1, \dots, k, \dots, n$ .

Si pensi ad una situazione in cui i flussi  $S_k$  ( $k=0, \dots, n$ ) confluiscono su un conto corrente bancario. Se  $D$  indica la disponibilità iniziale nel conto corrente, il saldo iniziale all'epoca 0 sarà uguale a

$$M_0 = D + S_0$$

Nell'ipotesi che sia applicato un tasso di interesse  $i$  se il saldo ad una determinata scadenza è positivo ed un tasso di interesse  $j$  ( $j > i$ ) se il saldo è negativo, si può calcolare il saldo ad una generica scadenza  $k$  come aggiornamento del saldo precedente (scadenza  $k-1$ ); l'operazione consiste nel contabilizzare l'importo  $S_k$  previsto alla scadenza  $k$  e contabilizzare gli interessi che il saldo precedente ha maturato. Gli interessi sono calcolati al tasso  $i$  o al tasso  $j$  a seconda del segno del saldo precedente. Il saldo all'epoca  $k$  viene perciò calcolato come

$$M_k = \begin{cases} M_{k-1}(1+i) + S_k & \text{se } M_{k-1} > 0 \\ M_{k-1}(1+j) + S_k & \text{se } M_{k-1} < 0 \end{cases}$$

In base al criterio T.R.M. (dal nome dei tre studiosi Teichroew, Robichek, Montalbano che lo hanno proposto) si valutano i progetti in base al saldo ottenuto alla scadenza ( $M_n$ ): si sceglie il progetto con il maggior saldo finale.

### 3.2 Uso della funzione SE

Si vuole impiegare Excel per risolvere problemi di scelta di progetti finanziari utilizzando la procedura T.R.M.

Si farà uso della funzione di Excel SE, che si presenta nella forma

$$SE(Test;se\_vero;se\_falso)$$

Gli argomenti della funzione hanno il seguente significato:

<i>Test</i>	È un valore o un'espressione che può dare come risultato VERO o FALSO.
<i>se_vero</i>	È il valore che viene restituito se <i>Test</i> è VERO.
<i>se_falso</i>	È il valore che viene restituito se <i>Test</i> è FALSO.

Si veda il seguente esempio.

#### **Esempio**

Valutare la convenienza dei progetti finanziari A e B, i cui movimenti di cassa sono di seguito riportati:

<b>PROGETTO A</b>		<b>PROGETTO B</b>	
<i>epoca</i>	<i>Importo</i>	<i>epoca</i>	<i>importo</i>
0	1500	0	900
1	-300	1	180
2	-1600	2	-600
3	590	3	-1300
4	1900	4	-60
5	200	5	1100
6	-1900	6	700

Per la valutazione si impieghi il criterio T.R.M. con tassi annuali del 9% sui saldi positivi e del 12% sui saldi negativi. Per entrambi i progetti la disponibilità iniziale è nulla ( $D=0$ ).

In un foglio di lavoro Excel, come quello visualizzato in figura 10, si procede come segue:

- Selezionare la cella B3, digitare il tasso di interesse sui saldi positivi, in questo caso: 0.09 e premere INVIO.
- Selezionare la cella B4, digitare il tasso di interesse sui saldi negativi, in questo caso: 0.12 e premere INVIO.
- Scrivere i movimenti di cassa del progetto A nelle celle da B8 a B14.

figura 10

	A	B	C	D	E	F
1	<b>Scelta di progetti - criterio T.R.M.</b>					
2						
3	tasso i (su saldi positivi)	0.09				
4	tasso j (su saldi negativi)	0.12				
5						
6	<b>scadenze</b>	<b>PROGETTO A</b>	<b>PROGETTO B</b>			
7		<b>movimenti di cassa</b>	<b>movimenti di cassa</b>			
8	0	1500	900			
9	1	-300	180			
10	2	-1600	-600			
11	3	590	-1300			
12	4	1900	-60			
13	5	200	1100			
14	6	-1900	700			
15						
16						
17	<b>scadenze</b>	<b>saldi progetto A</b>	<b>saldi progetto B</b>			
18						
19	0	1500.00	900.00			
20	1	1335.00	1161.00			
21	2	-144.85	665.49			
22	3	427.77	-574.62			
23	4	2366.27	-703.57			
24	5	2779.23	312.00			
25	6	<b>1129.36</b>	<b>1040.08</b>	<b>SALDO FINALE</b>		

- Scrivere i movimenti di cassa del progetto B nelle celle da C8 a C14.
  - Per calcolare i saldi del progetto A
    - Nella cella B19 scrivere la formula  
=B8
- e premere INVIO.

- Nella cella B20 scrivere la formula

$$= \text{SE}(B19>0;B19*(1+\$B\$3)+B9;B19*(1+\$B\$4)+B9)$$

e premere INVIO.

Utilizzando il linguaggio comune la formula significa:

”Se il valore della cella B19 (corrispondente al saldo  $M_0$  all’epoca zero) è positivo, allora esegui l’operazione  $M_0(1+i)+S_0$ , altrimenti esegui l’operazione:  $M_0(1+j)+S_0$ ”.

- Selezionare la cella B20.
- Scegliere il comando *Copia* dal menu *Modifica*.
- Selezionare le celle da B21 a B25.
- Scegliere il comando *Incolla* dal menu *Modifica*.  
Excel calcolerà automaticamente gli altri saldi del progetto A, modificando opportunamente la formula di calcolo del saldo all’epoca 1.
- Per calcolare i saldi del progetto B
  - Selezionare le celle da B19 a B25.
  - Scegliere il comando *Copia* dal menu *Modifica*.
  - Selezionare le celle da C19 a C25.
  - Scegliere il comando *Incolla* dal menu *Modifica*.  
Excel calcolerà automaticamente i saldi del progetto B, modificando opportunamente le formule del calcolo dei saldi del progetto A.

I valori dei saldi relativi ai due progetti si leggono nelle celle da B19 a B25 per il progetto A e nelle celle da C19 a C25 per il progetto B (vedere figura 10). Poiché A è caratterizzato da un saldo finale maggiore di B, esso verrà valutato come il progetto più conveniente.

## VALUTAZIONE DI RENDITE

Data una rendita, cioè una successione di importi monetari che si verificano in epoche determinate, si considerano in questa sede unicamente rendite periodiche (gli istanti temporali sono equidistanziati), costanti (l'importo periodico monetario (rata) è uguale in ciascuna epoca) e temporanee (la durata della rendita è finita).

### 1. MONTANTE DI RENDITE PERIODICHE POSTICIPATE

Si consideri una rendita annua posticipata di rata  $R$  e durata  $n$ . La rendita consiste nella riscossione (o pagamento) di una somma  $R$  alla fine di ogni anno, per  $n$  anni. Utilizzando il regime della capitalizzazione composta al tasso di interesse annuo  $i$  il montante  $M$  di tale rendita si calcola come

$$M = R \frac{(1+i)^n - 1}{i}$$

Per calcolare il montante della rendita si può usare la funzione finanziaria di Excel VAL.FUT. La funzione si presenta nella forma

$$VAL.FUT(tasso\_int;periodi;pagamento;valore\_attuale;tipo)$$

Gli argomenti della funzione hanno il seguente significato (con riferimento al problema della valutazione di una rendita):

<i>tasso_int</i>	Tasso di interesse relativo al periodo considerato.
<i>periodi</i>	Numero $n$ delle rate (corrispondente anche alla durata della rendita).
<i>pagamento</i>	Importo $R$ della rata riscossa (o pagata) in ciascun periodo.
<i>valore_attuale</i>	Rappresenta un flusso iniziale (riscosso o pagato all'epoca 0), il cui importo può anche essere diverso dall'importo delle rate.
<i>tipo</i>	Se <i>tipo</i> è uguale al numero 0 (oppure se l'argomento viene omissso) significa che si stanno considerando rate posticipate, altrimenti ( <i>tipo</i> uguale ad 1) si stanno considerando rate anticipate.

Qualora la rendita sia periodica, ma con un periodo diverso dall'anno (es. rendita semestrale), il tasso di interesse da inserire come argomento della funzione VAL.FUT dovrà essere il tasso effettivo riferito al periodo di tempo considerato (es. tasso effettivo semestrale).

Si fa notare che VAL.FUT restituisce un numero che ha segno opposto al segno di *pagamento*, per cui la funzione VAL.FUT restituisce un numero che è l'opposto del valore del montante cercato. Il seguente esempio può essere chiarificatore.

**Esempio (Calcolo del montante)**


Calcolare il montante di una rendita annua posticipata di L.5 milioni e durata 6 anni, al tasso di interesse del 4.4% annuo.

Per risolvere il problema si effettuano le seguenti operazioni, considerando un foglio di lavoro Excel come quello evidenziato in figura 11.

- Selezionare la cella B3, digitare l'importo della rata, cioè: 5000000 e premere INVIO.
- Selezionare la cella B4, digitare la durata della rendita, cioè: 6 e premere INVIO.
- Selezionare la cella B5, digitare il valore del tasso di interesse, cioè: 0.044 e premere INVIO.
- Selezionare la cella B7, digitare la formula per il calcolo del montante

$$= -\text{VAL.FUT}(B5;B4;B3)$$

figura 11



e premere INVIO. Si ottiene come risultato L. 33500102.

Si osservi che in alternativa all'impiego della funzione finanziaria di Excel VAL.FUT, l'ultima operazione descritta poteva essere sostituita dalla seguente:

- Selezionare la cella B7, digitare direttamente la formula per il calcolo del montante

$$=B3*((1+B5)^B4-1)/B5$$

e premere INVIO. Si ottiene ancora come risultato L. 33500102, ma in questo modo si sono impiegati molti più caratteri.

### ....usando i nomi delle celle

Vogliamo ora facilitare la lettura delle formule Excel usando nelle formule il nome assegnato a ciascuna cella. Per assegnare i nomi alle celle si può procedere in due modi:

i) (Uso della *Casella Nome* nella barra della formula). Si seleziona la cella che si vuole nominare. Si seleziona la *Casella Nome* nella barra della formula e si digita il nome che si vuole dare alla cella. Si preme INVIO.

ii) (Uso del comando *Nome* nel menu *Inserisci*). Si seleziona la cella che si vuole nominare. Dal menu *Inserisci* si sceglie *Nome* e quindi *Definisci*. Si apre la finestra di dialogo *Definisci nome*. Nella casella *Nomi nella cartella di lavoro* si scrive il nome che si vuole dare alla cella. Confermare con OK.

Riprendiamo il problema del calcolo del montante di una rendita, considerato nell'esempio precedente. Nel foglio di lavoro corrente usato nell'esempio effettuiamo le seguenti operazioni, al fine di assegnare un nome alle celle e poi utilizzare questi nomi nelle formule al posto dei riferimenti di cella (vedere figura 12)

- Selezionare la cella B3. Selezionare la *Casella Nome* nella barra della formula e digitare **rata**. Premere INVIO.
- Selezionare la cella B4. Selezionare la *Casella Nome* nella barra della formula e digitare **n**. Premere INVIO.
- Selezionare la cella B5. Selezionare la *Casella Nome* nella barra della formula e digitare **i**. Premere INVIO.
- Selezionare la cella B7, digitare la formula per il calcolo del montante

$$= -VAL.FUT(i;n;rata)$$

e premere INVIO. Si ottiene come risultato L. 33500102.


In alternativa all'impiego della funzione VAL.FUT, come mostrato in precedenza, l'ultima operazione descritta poteva essere sostituita dalla seguente:

- Selezionare la cella B7, digitare la formula per il calcolo del montante

$$= \text{rata} * ((1+i)^n - 1) / i$$

e premere invio. Si ottiene ancora come risultato L. 33500102.

figura 12



**Nota:** dopo aver nominato le celle, per sostituire direttamente nelle formule i riferimenti di cella con i nomi corrispondenti (senza riscrivere le formule), si può scegliere *Nome* dal menu *Inserisci* e quindi scegliere *Applica*. Si apre la finestra di dialogo *Applica nomi*, che elenca tutti i nomi che sono stati definiti. Selezionare i nomi che si desidera sostituire ai riferimenti di cella (per selezionare più di un nome tenere premuto il tasto MAIUSC mentre si seleziona ogni nome). Confermare con OK.


### ***Calcolo della rata, noto il montante***

*Considerando gli stessi dati del problema precedente (rendita annua posticipata con durata di 6 anni e tasso annuo di interesse del 4.4%) dire quale dovrebbe essere l'importo  $R$  della rata affinché il montante finale ammonti a L. 40 milioni.*

Per risolvere il problema del calcolo della rata si procede come segue:

- Effettuare una copia del foglio di lavoro corrente usato nell'esempio precedente (Posizionarsi nel foglio di lavoro corrente. Selezionare Sposta o copia foglio dal menu Modifica. Nella voce Prima di della finestra di dialogo Sposta o copia selezionare Foglio2. Selezionare l'opzione Crea una copia. Infine confermare con OK).
- Nel nuovo foglio di lavoro (Foglio1(2)) selezionare *Ricerca obiettivo* dal menu *Strumenti*.
- Si apre la finestra di dialogo *Ricerca obiettivo*. Nella casella *Imposta la cella* della finestra di dialogo scrivere B7.
- Nella casella *Al valore* digitare il valore del montante finale che si desidera ottenere, cioè: 40000000.
- Nella casella *Cambiando la cella* digitare B3 (il riferimento alla cella contenente l'importo della rata).
- Confermare con OK. Appare la finestra di dialogo *Stato ricerca obiettivo* che informa che è stata trovata una soluzione.
- Confermare con OK. Il risultato cercato comparirà nella cella B3. Si ottiene  $R=5970131$  (vedere figura 13).

figura 13



## 2. MONTANTE DI RENDITE PERIODICHE ANTICIPATE

Si consideri ora una rendita annua in cui la riscossione (o il pagamento) degli importi monetari  $R$  avviene in modo anticipato all'inizio di ogni anno, per una durata complessiva di  $n$  anni.

Anche per calcolare il montante di una rendita anticipata si può usare la funzione finanziaria

$$VAL.FUT(tasso\_int;periodi;pagamento;valore\_attuale;tipo)$$

dove gli argomenti della funzione hanno lo stesso significato visto nel paragrafo precedente, ma con l'avvertenza che nel caso di rendite anticipate l'argomento *tipo* dovrà essere posto uguale ad 1.

### **Esempio**

*Calcolare il montante di una rendita che prevede la riscossione di L 2 milioni all'inizio di ogni anno, per 10 anni, qualora il tasso di interesse sia uguale al 5% annuo.*

Per calcolare il montante richiesto:

- Selezionare una qualsiasi cella di un foglio di lavoro Excel,

- Digitare direttamente la formula per il calcolo del montante

$$= - VAL.FUT(0.05;10;2000000;;1)$$

e premere il tasto INVIO. Si ottiene come risultato un montante uguale a L. 26,413,574.32.

Da notare nella funzione VAL.FUT il punto e virgola aggiuntivo che agisce da separatore per l'argomento della funzione *val\_attuale* non impiegato.

### 3. VALORE ATTUALE DI RENDITE PERIODICHE

Il calcolo del valore attuale di una rendita periodica di rata  $R$  e durata  $n$  può calcolarsi, sia nel caso di rendite posticipate che nel caso di rendite anticipate, utilizzando la funzione finanziaria VA di Excel.

Si è già avuto modo di descrivere la funzione VA in precedenza (nel paragrafo 1.3 del capitolo dedicato alla scelta di progetti finanziari). Riassumiamo in questo paragrafo la descrizione degli argomenti della funzione VA considerando il loro significato in problemi di valutazione di rendite:

$$VA(\text{tasso\_int};\text{periodi};\text{pagamento};\text{valore\_futuro};\text{tipo})$$

dove

*tasso\_int.*

Tasso di interesse relativo al periodo considerato.

*periodi*

Numero  $n$  delle rate (durata della rendita).

*pagamento*

Importo  $R$  della rata riscossa (o pagata) in ciascun periodo.

*valore\_futuro*

Argomento opzionale; pagamento addizionale  $F$  effettuato nell'ultima epoca, che si aggiunge all'ultimo pagamento.

*tipo*

Se *tipo* è uguale a 0 (oppure omissis) indica che la rendita è posticipata; se è uguale ad 1 indica che la rendita è anticipata.

Si ricorda che la funzione VA fornisce un valore di segno opposto rispetto al segno della rata  $R$ .

#### **Esempio**

*Calcolare il valore attuale di una rendita che prevede la riscossione mensile, in via anticipata, di L. 500000, per 4 anni. Si effettuino i conteggi considerando un tasso di interesse del 12% annuo in regime di capitalizzazione composta.*

Si procede come segue, al fine di ottenere un foglio di lavoro come quello visualizzato in figura 14.

- Selezionare la cella B3, digitare l'importo della rata mensile cioè: 500000 e premere INVIO.
- Selezionare la cella B4, digitare la durata della rendita in anni cioè: 4 e premere INVIO.
- Selezionare la cella B5, digitare il valore del tasso annuo di interesse cioè: 0.12 e premere INVIO.
- Selezionare la cella B7, digitare la formula per il calcolo della durata della rendita in mesi:

$$=B4*12$$

e premere INVIO. Si ottiene come risultato 48.

- Selezionare la cella B8, digitare la formula per il calcolo del tasso di interesse mensile equivalente al tasso annuo di interesse del 12%

$$=(1+B5)^(1/12)-1$$

e premere INVIO. Si ottiene come risultato 0.009488793.

- Selezionare la cella B10, digitare la formula per il calcolo del valore attuale della rendita mensile anticipata

$$=-VA(B8;B7;B3;;1)$$

e premere INVIO. Si ottiene un valore attuale di L 19388157.

figura 14

	A	B	C	D
1	<b>Valore attuale di una rendita mensile anticipata</b>			
2				
3	importo (mensile) della rata	500000		
4	durata della rendita (in anni)	4		
5	tasso annuo di interesse	0.12		
6				
7	durata della rendita (in mesi)	48		
8	tasso mensile equivalente	0.009488793	$i_{12} = (1+B5)^{(1/12)} - 1$	
9				
10	Valore attuale	19388157		
11				
12				
13				
14				
15				
16				
17				
18				

## PIANO DI AMMORTAMENTO DI UN DEBITO

Si consideri il problema di redigere il piano di ammortamento di un debito che specifica i tempi di rimborso del capitale ed il pagamento degli interessi sul prestito.

Si indichi con:

$S$	ammontare del debito
$n$	durata del prestito (numero di anni)
$k$	generico anno ( $k = 0, \dots, n$ )
$R_k$	rata di ammortamento in $k$
$C_k$	quota capitale in $k$
$I_k$	quota interesse in $k$
$D_k$	debito residuo in $k$
$E_k$	debito estinto in $k$
$i$	tasso di interesse annuo

Si consideri la situazione in cui i pagamenti vengono corrisposti posticipatamente al termine di ciascuno degli  $n$  anni. Fra le grandezze del piano di ammortamento valgono le seguenti relazioni:

$$\begin{aligned}R_k &= C_k + I_k \\I_k &= iD_{k-1} \\D_k &= D_{k-1} - C_k \\E_k &= E_{k-1} + C_k\end{aligned}$$

valgono inoltre le seguenti condizioni di chiusura

$$\begin{aligned}\sum_{k=1}^n C_k &= S \\ \sum_{k=1}^n R_k (1+i)^{-k} &= S\end{aligned}$$

## 1. AMMORTAMENTO ITALIANO

In un ammortamento di tipo italiano, caratterizzato da quote di capitale costanti, l'ammontare di ciascuna quota capitale è dato da

$$C_k = \frac{S}{n} \quad k = 1, 2, \dots, n$$

### *Esempio*

*Si costruisca, utilizzando Excel, il piano di ammortamento completo di un prestito di 30000 Euro al 5.5% annuo in 6 anni, con quote di capitale costanti.*

Si procede come segue:

- In un foglio di lavoro Excel, come quello visualizzato in figura 15, digitare i dati del problema nelle celle da B3 a B5:
  - Nella cella B3 digitare l'ammontare del debito: 30000 e premere INVIO.
  - Nella cella B4 digitare la durata del prestito: 6 e premere INVIO.
  - Nella cella B5 digitare il tasso annuo di interesse: 0.055 e premere INVIO
- Nelle celle da B10 a B16 digitare i numeri da 0 a 6, corrispondenti agli anni di rimborso.
- Nella cella F10 del debito residuo iniziale digitare il riferimento alla cella contenente l'ammontare del debito, cioè:  
=B3  
e premere INVIO.
- Nella cella G10 corrispondente al debito estinto iniziale digitare 0 e premere INVIO.
- Per calcolare i pagamenti relativi al primo anno  
(vedere la figura 15 per le formule e la figura 16 per i risultati)

figura 15

Microsoft Excel - Ammortamento-0bis.xls

File Modifica Visualizza Inserisci Formato Strumenti Dati Finestra ?

Salva con nome...

Times New Roman 14

A1 = Ammortamento con quote di capitale costanti

Ammortamento con quote di capitale costanti						
1	<b>Ammortamento con quote di capitale costanti</b>					
2						
3	<i>debito (S)</i>	30000				
4	<i>numero anni (n)</i>	6				
5	<i>tasso di interesse</i>	0.055				
6						
7	<b>Piano di ammortamento</b>					
8						
9		<i>quota capitale (Ck)</i>	<i>quota interessi (Ik)</i>	<i>rata (Rk)</i>	<i>Debito residuo (Dk)</i>	<i>Debito estinto (Ek)</i>
10	anno (k)					
11	0				=B3	0
12	1	=B3/\$B\$4	=F10*\$B\$5	=C11+D11	=F10-C11	=G10+C11
13	2					
14	3					
15	4					
16	5					
17	6					

Foglio1 / Foglio2 / Foglio3 / Pronto NUM

figura 16

Microsoft Excel - Ammortamento-0.xls

File Modifica Visualizza Inserisci Formato Strumenti Dati Finestra ?

Salva con nome...

Times New Roman 14

A1 = Ammortamento con quote di capitale costanti

Ammortamento con quote di capitale costanti						
1	<b>Ammortamento con quote di capitale costanti</b>					
2						
3	<i>debito (S)</i>	30000				
4	<i>numero anni (n)</i>	6				
5	<i>tasso di interesse</i>	0.055				
6						
7	<b>Piano di ammortamento</b>					
8						
9		<i>quota capitale (Ck)</i>	<i>quota interessi (Ik)</i>	<i>rata (Rk)</i>	<i>Debito residuo (Dk)</i>	<i>Debito estinto (Ek)</i>
10	anno (k)					
11	0				30000	0
12	1	5000	1650	6650	25000	5000
13	2					
14	3					
15	4					
16	5					
17	6					

Foglio1 / Foglio2 / Foglio3 / Pronto NUM

- Selezionare la cella C11, digitare la formula per il calcolo della quota capitale:  

$$=B\$3/B\$4$$
 e premere INVIO. Si ottiene come risultato 5000.
  - Selezionare la cella D11, digitare la formula per il calcolo della prima quota interesse:  

$$=F10*B\$5$$
 e premere INVIO. Si ottiene come risultato 1650.
  - Selezionare la cella E11, digitare la formula per il calcolo della prima rata di ammortamento:  

$$=C11+D11$$
 e premere INVIO. Si ottiene come risultato 6650.
  - Selezionare la cella F11, digitare la formula per il calcolo del debito residuo al primo anno:  

$$=F10-C11$$
 e premere INVIO. Si ottiene come risultato 25000.
  - Selezionare la cella G11, digitare la formula per il calcolo del debito estinto al primo anno:  

$$=G10+C11$$
 e premere INVIO. Si ottiene come risultato 5000.
- Per calcolare i pagamenti degli anni successivi al primo
- Selezionare l'intervallo di celle C11:G11.
  - Scegliere il comando *Copia* dal menu *Modifica*.
  - Selezionare l'intervallo con 5 righe e 5 colonne C12:G16 (vedere figura 17).

figura 17

The screenshot shows an Excel spreadsheet titled "Ammortamento-1b.xls". The data is organized as follows:

Ammortamento con quote di capitale costanti						
debito (S)	30000					
numero anni (n)	6					
tasso di interesse	0.055					
<b>Piano di ammortamento</b>						
anno (k)	quota capitale (C <sub>k</sub> )	quota interessi (I <sub>k</sub> )	rata (R <sub>k</sub> )	Debito residuo (D <sub>k</sub> )	Debito estinto (E <sub>k</sub> )	
0				30000	0	
1	5000	1650	6650	25000	5000	
2						
3						
4						
5						
6						

- Scegliere il comando *Incolla* dal menu *Modifica*.  
Excel calcolerà automaticamente tutti i pagamenti relativi agli anni successivi al primo, modificando opportunamente le formule del calcolo dei primi pagamenti.  
Le figure seguenti riportano il piano di ammortamento completo, evidenziando sia le formule (figura 18) che i risultati (figura 19).

figura 18

Ammortamento con quote di capitale costanti						
	<i>debito (S)</i>	30000				
	<i>numero anni (n)</i>	6				
	<i>tasso di interesse</i>	0.055				
<b>Piano di ammortamento</b>						
anno (k)	quota capitale (Ck)	quota interessi (Ik)	rata (Rk)	Debito residuo (Dk)	Debito estinto (Ek)	
0				=B3		0
1	=B3/\$B\$4	=F10*\$B\$5	=C11+D11	=F10-C11	=G10+C11	
2	=B3/\$B\$4	=F11*\$B\$5	=C12+D12	=F11-C12	=G11+C12	
3	=B3/\$B\$4	=F12*\$B\$5	=C13+D13	=F12-C13	=G12+C13	
4	=B3/\$B\$4	=F13*\$B\$5	=C14+D14	=F13-C14	=G13+C14	
5	=B3/\$B\$4	=F14*\$B\$5	=C15+D15	=F14-C15	=G14+C15	
6	=B3/\$B\$4	=F15*\$B\$5	=C16+D16	=F15-C16	=G15+C16	

figura 19

Ammortamento con quote di capitale costanti						
	<i>debito (S)</i>	30000				
	<i>numero anni (n)</i>	6				
	<i>tasso di interesse</i>	0.055				
<b>Piano di ammortamento</b>						
anno (k)	quota capitale (Ck)	quota interessi (Ik)	rata (Rk)	Debito residuo (Dk)	Debito estinto (Ek)	
0				30000		0
1	5000	1650	6650	25000	5000	
2	5000	1375	6375	20000	10000	
3	5000	1100	6100	15000	15000	
4	5000	825	5825	10000	20000	
5	5000	550	5550	5000	25000	
6	5000	275	5275	0	30000	

## 2. AMMORTAMENTO FRANCESE

In un ammortamento di tipo francese caratterizzato da rate di ammortamento costanti l'ammontare  $R$  di ciascuna rata è calcolata come

$$R = S \frac{i}{1 - (1 + i)^{-n}}$$

Per calcolare l'importo della rata si può usare la funzione finanziaria di Excel RATA. La funzione si presenta nella forma:

$RATA(tasso\_int; periodi; val\_attuale; val\_futuro; tipo)$

Gli argomenti della funzione hanno il seguente significato:

<i>tasso_int</i>	Tasso di interesse relativo al periodo considerato.
<i>periodi</i>	Numero $n$ delle rate di ammortamento.
<i>valore_attuale</i>	Importo del mutuo da ammortizzare.
<i>valore_futuro</i>	È il valore residuo del capitale, dopo l'ultimo pagamento. Se omesso Excel lo considera pari a zero.
<i>tipo</i>	È il tipo di convenzione usata per i pagamenti (0 = Posticipato, 1 = Anticipato). Se l'argomento viene omesso Excel lo considera uguale a zero.

La funzione restituisce l'importo della rata necessario per ammortizzare un debito (secondo un piano di ammortamento francese), noti il tasso di remunerazione e la durata del prestito. La funzione rata restituisce un risultato negativo in quanto il suo importo rappresenta un costo, dal punto di vista del debitore.

### **Esempio**

*Redigere il piano di ammortamento di un debito di L. 64000 Euro estinguibile in 5 anni al tasso del 7%, nell'ipotesi di un ammortamento a rate costanti.*

Si procede come segue:

- In un foglio di lavoro Excel, come quello visualizzato in figura 20, digitare i dati del problema nelle celle da B3 a B5:
  - Nella cella B3 digitare l'ammontare del debito: 64000 e premere INVIO.
  - Nella cella B4 digitare la durata del prestito: 5 e premere INVIO.
  - Nella cella B5 digitare il tasso annuo di interesse: 0.07 e premere INVIO.

- Selezionare la cella B7 e digitare la formula per il calcolo della rata

=-RATA(B5;B4;B3)

e premere INVIO. Si ottiene come risultato 15609.

- Nelle celle da B11 a B16 digitare i numeri da 0 a 5, corrispondenti agli anni di rimborso.

- Selezionare la cella F11 del debito residuo iniziale digitare il riferimento alla cella contenente l'ammontare del debito, cioè:

=B3

e premere INVIO.

- Selezionare la cella G11 corrispondente al debito estinto iniziale, digitare 0 e premere INVIO.

- Per calcolare i pagamenti relativi al primo anno  
(vedere la figura 20 per le formule e la figura 21 per i risultati)

- Selezionare la cella E12, digitare il riferimento alla cella contenente l'importo della rata, cioè:

=\$B\$7

e premere INVIO. Si ottiene come risultato 15609.

- Selezionare la cella D12, digitare la formula per il calcolo della prima quota interesse:

=\$B\$5\*F11

e premere INVIO. Si ottiene come risultato 4480.

- Selezionare la cella C12, digitare la formula per il calcolo della prima quota capitale:

=E12-D12

e premere INVIO. Si ottiene come risultato 11129.

- Selezionare la cella F12, digitare la formula per il calcolo del debito residuo al primo anno:

=F11-C12

e premere INVIO. Si ottiene come risultato 52871.

- Selezionare la cella G12, digitare la formula per il calcolo del debito estinto al primo anno:

$$= G11+C12$$

e premere INVIO. Si ottiene come risultato 11129.

figura 20

Ammortamento con rate costanti						
debito (S)	64000					
numero anni (n)	5					
tasso di interesse	0.07					
Rata (R)	=-RATA(B5;B4;B3)					
Piano di ammortamento						
anno (k)	quota capitale (C <sub>k</sub> )	quota interessi (I <sub>k</sub> )	rata (R <sub>k</sub> )	Debito residuo (D <sub>k</sub> )	Debito estinto (E <sub>k</sub> )	
0				=B3	0	
1	=E12-D12	=\$B\$5*F11	=\$B\$7	=F11-C12	=G11+C12	
2						
3						
4						
5						

figura 21

Ammortamento con rate costanti						
debito (S)	64000					
numero anni (n)	5					
tasso di interesse	0.07					
Rata (R)	15609					
Piano di ammortamento						
anno (k)	quota capitale (C <sub>k</sub> )	quota interessi (I <sub>k</sub> )	rata (R <sub>k</sub> )	Debito residuo (D <sub>k</sub> )	Debito estinto (E <sub>k</sub> )	
0				64000	0	
1	11129	4480	15609	52871	11129	
2						
3						
4						
5						

- Per calcolare i pagamenti degli anni successivi al primo
  - Selezionare l'intervallo di celle C12:G12.
  - Scegliere il comando *Copia* dal menu *Modifica*.
  - Selezionare l'intervallo con 4 righe e 5 colonne C13:G16 Scegliere il comando *Incolla* dal menu *Modifica*.


Excel calcolerà automaticamente tutti i pagamenti relativi agli anni successivi al primo, modificando opportunamente le formule del calcolo dei primi pagamenti. Le figure seguenti riportano il piano di ammortamento completo, evidenziando sia le formule (figure 22) che i risultati (figura 23).

figura 22

The screenshot shows an Excel spreadsheet titled "Ammortamentofrancese1.xls". The spreadsheet contains the following data and formulas:

Ammortamento con rate costanti						
debito (S)	64000					
numero anni (n)	5					
tasso di interesse	0.07					
Rata (R)	=-RATA(B5;B4;B3)					
Piano di ammortamento						
anno (k)	quota capitale (C <sub>k</sub> )	quota interessi (I <sub>k</sub> )	rata (R <sub>k</sub> )	Debito residuo (D <sub>k</sub> )	Debito estinto (E <sub>k</sub> )	
0				=B3		0
1	=E12-D12	=\$B\$5*F11	=\$B\$7	=F11-C12	=G11+C12	
2	=E13-D13	=\$B\$5*F12	=\$B\$7	=F12-C13	=G12+C13	
3	=E14-D14	=\$B\$5*F13	=\$B\$7	=F13-C14	=G13+C14	
4	=E15-D15	=\$B\$5*F14	=\$B\$7	=F14-C15	=G14+C15	
5	=E16-D16	=\$B\$5*F15	=\$B\$7	=F15-C16	=G15+C16	

figura 23



The screenshot shows an Excel spreadsheet titled "Ammortamento con rate costanti". The input parameters are:

- debito (S): 64000
- numero anni (n): 5
- tasso di interesse: 0.07
- Rata (R): 15609

The amortization table is as follows:

anno (k)	quota capitale (C <sub>k</sub> )	quota interessi (I <sub>k</sub> )	rata (R <sub>k</sub> )	Debito residuo (D <sub>k</sub> )	Debito estinto (E <sub>k</sub> )
0				64000	0
1	11129	4480	15609	52871	11129
2	11908	3701	15609	40963	23037
3	12742	2867	15609	28221	35779
4	13634	1975	15609	14588	49412
5	14588	1021	15609	0	64000

**NOTA**

Per calcolare la quota interesse  $I_k$ , e la quota capitale  $C_k$  relative ad un determinato anno  $k$ , è possibile usare le funzioni di Excel INTERESSI e P.RATA. Le funzioni si presentano nella forma:

$$\text{INTERESSI}(\text{tasso\_int}; \text{periodo}; \text{periodi}; \text{val\_attuale}; \text{val\_futuro}; \text{tipo})$$

$$\text{P.RATA}(\text{tasso\_int}; \text{periodo}; \text{periodi}; \text{val\_attuale}; \text{val\_futuro}; \text{tipo})$$

Gli argomenti della funzione *tasso\_int*, *periodi*, *val\_attuale*, *val\_futuro* e *tipo* hanno lo stesso significato di quelli della funzione RATA. In aggiunta troviamo l'argomento *periodo*, che rappresenta l'epoca  $k$  di calcolo della quota interesse e della quota capitale. Come la funzione RATA, anche le funzioni INTERESSI e P.RATA danno come risultato un numero negativo.

Nell'esempio precedente, se selezioniamo la cella E3, digitiamo la formula  
 $= -\text{INTERESSI} (B5;2;B4;B3)$

e premiamo INVIO, otteniamo la quota interesse del secondo anno, cioè 3701.

Se selezioniamo la cella E4, digitiamo la formula  
 $= -\text{P.RATA}(B5;2;B4;B3)$

e premiamo INVIO, otteniamo la quota capitale del secondo anno, cioè 11908.

## ANALISI PER IPOTESI: USO DI TABELLE DI DATI

Excel fornisce alcune funzioni avanzate per eseguire analisi per ipotesi, al fine di verificare l'impatto sul modello decisionale di cambiamenti nel valore di alcune variabili.

Le *tabelle di dati* possono essere utili per riassumere le conseguenze di variazioni successive di una sola variabile (tabelle con una variabile di input) oppure per presentare le interazioni fra due variabili (tabelle con due variabili di input). Per costruire una tabella di dati si usa il comando *Tabella* nel menu *Dati*. Vediamo i seguenti esempi.

### **1. Esempio: una variabile di input**

*Si voglia acquistare un prodotto a rate, tramite il ricorso ad un mutuo di L. 16 milioni, rimborsabile in 8 anni con un ammortamento a rate costanti. Analizzare il cambiamento nel valore della rata al variare del tasso di interesse sul prestito dal 5% al 10% (con intervalli di mezzo punto percentuale).*

In un foglio di lavoro Excel si proceda come segue:

- Selezionare la cella B3, digitare l'importo del mutuo: 16000000 e premere INVIO.
  
- Inserire nell'intervallo di celle A7:A17 i valori del tasso di interesse che si desidera valutare (l'intervallo A7:A17 sarà considerato l'*intervallo di input*). Per inserire i valori del tasso di interesse variabili dal 5% al 10% con intervalli di mezzo punto, si può utilizzare il riempimento automatico:
  - selezionare la cella A7, digitare 0.05 e premere INVIO
  - selezionare la cella A8, digitare 0.055 e premere INVIO
  - selezionare sia la cella A7 che A8 e trascinare il quadratino di riempimento fino alla cella A17. Excel inserisce automaticamente i valori del tasso di interesse con intervalli di mezzo punto;
  - per visualizzare i valori del tasso di interesse in formato percentuale, selezionare l'intervallo di celle A7:A17, selezionare *Celle* dal menu *Formato*. Nella finestra di dialogo *Formato celle*, nella casella di riepilogo *Categoria* selezionare la voce *Percentuale* e confermare con OK.
  
- Selezionare la cella B6 e digitare la formula per il calcolo della rata di ammortamento
$$= -RATA(A6;8;B3)$$
Premere INVIO. Si ottiene come risultato L. 2000000.

Si osservi che la formula fa riferimento alla cella A6, che è vuota (la cella A6 agisce da segnaposto; tramite la cella A6 Excel andrà poi ad esaminare i valori compresi nell'intervallo di input).

- Selezionare l'intervallo di celle A6:B17.
- Selezionare *Tabella* dal menu *Dati*. Appare la finestra di dialogo Tabella (vedere figura 24). Nella casella di testo *Cella di input per colonna* digitare  $\$A\$6$ . Confermare con OK. Excel inserisce i risultati della formula per il calcolo della rata nelle celle B7:B17 (*intervallo di risultati*). Vedere figura 25.

[Si osservi che Excel inserisce in ogni cella dell'intervallo dei risultati B7:B17 la formula matrice  $\{=TABELLA(;A6)\}$ . Tale formula calcola i risultati della funzione RATA utilizzando come valore del tasso di interesse i valori considerati nell'intervallo di input].

figura 24




figura 25

Uso di una Tabella di dati (una variabile di input)	
mutuo	Importo rata
16000000	2000000
5.0%	2475549
5.5%	2525824
6.0%	2576575
6.5%	2627797
7.0%	2679484
7.5%	2731632
8.0%	2784236
8.5%	2837290
9.0%	2890790
9.5%	2944730
10.0%	2999104

## 2. Esempio: due variabili di input

Si consideri la stessa situazione dell'esempio precedente (mutuo di L. 16 milioni, rimborsabile con un ammortamento a rate costanti). Analizzare il cambiamento nel valore della rata al variare sia del tasso di interesse sul prestito che della durata del prestito. Si considerino tassi di interesse variabili dal 5% al 10% (con intervalli di mezzo punto percentuale) e si considerino durate di 3,6,9,12,15,18 anni.

In un foglio di lavoro Excel si proceda come segue:

- Selezionare la cella B3, digitare l'importo del mutuo: 16000000 e premere INVIO.
- Inserire nell'intervallo di celle B7:B17 i valori del tasso di interesse che si desidera valutare (l'intervallo B7:B17 sarà considerato un *intervallo di input*). Per inserire i valori del tasso di interesse variabili dal 5% al 10% con intervalli di mezzo punto, si può utilizzare il riempimento automatico, come visto in precedenza:
  - selezionare la cella B7, digitare 0.05 e premere INVIO
  - selezionare la cella B8, digitare 0.055 e premere INVIO

- selezionare sia la cella B7 che B8 e trascinare il quadratino di riempimento fino alla cella B17. Excel inserisce automaticamente i valori del tasso di interesse con intervalli di mezzo punto;
  - per visualizzare i valori del tasso di interesse in formato percentuale, selezionare l'intervallo di celle B7:B17, selezionare *Celle* dal menu *Formato*. Nella finestra di dialogo *Formato celle*, nella casella di riepilogo *Categoria* selezionare la voce *Percentuale* e confermare con OK.
- Inserire nell'intervallo di celle C6:H6 le durate del prestito che si desidera valutare (anche l'intervallo C6:H6 sarà considerato un *intervallo di input*). Si può utilizzare il riempimento automatico:
- selezionare la cella C6, digitare 3 e premere INVIO
  - selezionare la cella D6, digitare 6 e premere INVIO
  - selezionare sia la cella C6 che D6 e trascinare il quadratino di riempimento fino alla cella H6. Excel inserisce automaticamente i valori delle durate del prestito.
- Selezionare la cella B6 (la cella di output deve trovarsi nell'intersezione tra la riga e la colonna dei due intervalli di input) e digitare la formula per il calcolo della rata di ammortamento
- $$= -RATA(A6;B5;B3)$$
- Premere INVIO. Si ottiene come risultato #DIV/0!, ma non dobbiamo preoccuparci, ciò è dovuto alle due celle vuote A6 e B5.
- Selezionare l'intervallo di celle B6:H17.
- Selezionare *Tabella* dal menu *Dati*. Appare la finestra di dialogo *Tabella*. Nella casella di testo *Cella di input per riga* digitare \$B\$5. Nella casella di testo *Cella di input per colonna* digitare \$A\$6. Confermare con OK. Excel inserisce i risultati della formula per il calcolo della rata nell'intervallo di celle C7:H17 (vedere figura 26). Si osservi che Excel inserisce in ogni cella dell'intervallo dei risultati C7:H17 la formula matrice {=TABELLA(;B5;A6)}.

figura 26

Microsoft Excel - Tabelladati2.xls

File Modifica Visualizza Inserisci Formato Strumenti Dati Finestra ?

Salva con nome... 100%

Times New Roman 14

A1 = Uso di una Tabella di dati (due variabili di input)

	A	B	C	D	E	F	G	H
1	<b>Uso di una Tabella di dati (due variabili di input)</b>							
2								
3	mutuo	16000000						
4								
5			<b>Durata (anni)</b>					
6		#DIV/0!	3	6	9	12	15	18
7		5.0%	5875337	3152279	2251041	1805207	1541477	1368740
8		5.5%	5930465	3202863	2301431	1856468	1594010	1422719
9		6.0%	5985757	3253802	2352356	1908432	1647404	1477705
10		6.5%	6041211	3305093	2403809	1961091	1701645	1533674
11	Tassi di interesse	7.0%	6096827	3356733	2455784	2014432	1756714	1590602
12		7.5%	6152602	3408718	2508275	2068445	1812596	1648463
13		8.0%	6208536	3461046	2561275	2123120	1869273	1707234
14		8.5%	6264628	3513713	2614780	2178446	1926727	1766887
15		9.0%	6320876	3566717	2668781	2234411	1984942	1827397
16		9.5%	6377279	3620053	2723273	2291003	2043899	1888738
17		10.0%	6433837	3673718	2778249	2348213	2103580	1950884
18								
19								
20								
21								
22								
23								

Foglio1 / Foglio2 / Foglio3

Pronto NUM

## BIBLIOGRAFIA

Basso A., Pianca P. *Appunti di matematica finanziaria*, CEDAM, Padova 2000.

Dodge M., Stinson C. *Guida all'uso Microsoft Excel 2000*, Mondadori Informatica, 1999.